



UNIVERSIDAD CATÓLICA DE SALTA
Licenciatura en Ciencia de Datos

Trabajo Práctico N° 1

Análisis de algoritmos computacionales

Grupo 5

Integrantes:

Baldoma, Fatima
Fagnani, Gianella Maria
Luna, Facundo Gaston
Marquetti, Marcelo Sergio
Semino Giaccone, Esteban David
Vidal Bernal, America

Profesor: Ing. Guillermo Saravia Sanchez

Fecha: 17 de agosto de 2026

Índice

Aclaraciones previas	3
Introducción	3
1. Lectura de dos valores y determinación del mayor	4
1.1. Algoritmo principal	4
2. Lectura de tres valores: mayor, menor y validación	5
2.1. Algoritmo principal	5
3. Sumatoria de números enteros entre 1 y 10	7
3.1. Algoritmo principal	7
4. Hipotenusa de un triángulo rectángulo	8
4.1. Algoritmo principal	8
5. Área y volumen de un cilindro	9
5.1. Algoritmo principal	9
6. Determinar si un número es par o impar	10
6.1. Algoritmo principal	10
7. Conversión de calificaciones numéricas a letras	11
7.1. Algoritmo principal (Si-Entonces)	11
8. Ordenar dos números de menor a mayor	12
8.1. Algoritmo principal	12
9. Determinar si un número es primo	13
9.1. Algoritmo principal	13
10. Corrección de errores: cálculo de estacionamiento	14
10.1. Algoritmo corregido	14
11. Conversión de velocidad: km/h a m/s	16

11.1. Algoritmo principal	16
12.Cálculo del promedio de notas	17
12.1. Algoritmo principal	17
13.Nómina de obreros - Constructora VILAZ S.A.	18
13.1. Algoritmo principal	18
14.Simulación de caja registradora	19
14.1. Algoritmo principal	19
15.Conversión de días a años, meses, semanas y días	21
15.1. Algoritmo principal (Mientras)	21

Aclaraciones previas

Los algoritmos presentados en este trabajo fueron desarrollados en pseudocódigo, utilizando la sintaxis del software PSeInt. Se optó por esta herramienta con el objetivo de centrar el análisis en la lógica de resolución de cada problema, evitando que aspectos particulares de la sintaxis de un lenguaje de programación específico interfieran en la evaluación del razonamiento algorítmico —objetivo central de esta etapa inicial de la carrera.

Adicionalmente, el código fuente completo de cada ejercicio se encuentra disponible en el siguiente repositorio de GitHub, con el fin de facilitar su verificación y prueba en el entorno de PSeInt. Cada ejercicio corresponde a un archivo `.psc` independiente:

<https://github.com/Facugl/data-science-coursework/tree/main/primer-anio/analisis-algoritmos/tp-1/ejercicios>

Introducción

El presente trabajo práctico consiste en el desarrollo de algoritmos utilizando pseudocódigo (PSeInt), abordando estructuras secuenciales, condicionales y repetitivas, con el objetivo de afianzar los conceptos fundamentales de la lógica de programación vistos en la Unidad 1 de la materia.

1. Lectura de dos valores y determinación del mayor

Desarrolle un algoritmo que permita leer dos valores distintos, determinar cuál de los dos valores es el mayor.

1.1. Algoritmo principal

```
1 Algoritmo MayorDeDos
2   Definir a, b Como Real
3
4   Escribir "Ingrese el primer valor:"
5   Leer a
6
7   Repetir
8     Escribir "Ingrese el segundo valor (distinto del
9       primero):"
10    Leer b
11    Si b = a Entonces
12      Escribir "Error: el valor debe ser distinto al
13        primero. Intente nuevamente."
14    FinSi
15  Hasta Que b <> a
16
17  Si a > b Entonces
18    Escribir "El mayor es: ", a
19  SiNo
20    Escribir "El mayor es: ", b
21  FinSi
22 FinAlgoritmo
```

Algoritmo 1: Determinar el mayor de dos valores

[↑ Volver al índice](#)

2. Lectura de tres valores: mayor, menor y validación

Desarrolle un algoritmo que permita leer tres valores y almacenarlos en las variables A, B y C respectivamente. El algoritmo debe imprimir cuál es el mayor y cuál es el menor. Recuerde constatar que los tres valores introducidos por el teclado sean valores distintos. Presente un mensaje de alerta en caso de que se detecte la introducción de valores iguales.

2.1. Algoritmo principal

```
1  Algoritmo MayorDeTres
2      Definir A, B, C Como Real
3
4      Escribir "Ingrese el primer valor:"
5      Leer A
6
7      Repetir
8          Escribir "Ingrese el segundo valor (distinto del
          primero):"
9          Leer B
10         Si B = A Entonces
11             Escribir "Error: el valor debe ser distinto al
              primero. Intente nuevamente."
12         FinSi
13     Hasta Que B <> A
14
15     Repetir
16         Escribir "Ingrese el tercer valor (distinto del primero
          y el segundo):"
17         Leer C
18         Si C = A O C = B Entonces
19             Escribir "Error: el valor debe ser distinto al
              primero y el segundo. Intente nuevamente."
20         FinSi
21     Hasta Que C <> A Y C <> B
22
23     Si A > B Entonces
24         Si A > C Entonces
25             Escribir "El mayor es: ", A
26         SiNo
27             Escribir "El mayor es: ", C
28         FinSi
29     SiNo
30         Si B > C Entonces
```

```
31      Escribir "El mayor es: ", B
32      SiNo
33          Escribir "El mayor es: ", C
34      FinSi
35  FinSi
36
37  Si A < B Entonces
38      Si A < C Entonces
39          Escribir "El menor es: ", A
40      SiNo
41          Escribir "El menor es: ", C
42      FinSi
43  SiNo
44      Si B < C Entonces
45          Escribir "El menor es: ", B
46      SiNo
47          Escribir "El menor es: ", C
48      FinSi
49  FinSi
50
51 FinAlgoritmo
```

Algoritmo 2: Mayor y menor de tres valores con validación

[↑ Volver al índice](#)

3. Sumatoria de números enteros entre 1 y 10

Desarrolle un algoritmo que realice la sumatoria de los números enteros comprendidos entre el 1 y el 10, es decir, $1 + 2 + 3 + \dots + 10$.

3.1. Algoritmo principal

```
1 Algoritmo SumatoriaEnterosEntre1Y10
2   Definir suma Como Real
3   suma <- 0
4
5   Para i <- 1 Hasta 10 Con Paso 1 Hacer
6       suma <- suma + i
7   Fin Para
8
9   Escribir "La suma de los enteros es: ", suma
10 FinAlgoritmo
```

Algoritmo 3: Sumatoria del 1 al 10

[↑ Volver al índice](#)

4. Hipotenusa de un triángulo rectángulo

Determinar la hipotenusa de un triángulo rectángulo conocidas las longitudes de sus dos catetos. Desarrolle el algoritmo correspondiente.

4.1. Algoritmo principal

```
1 Algoritmo HipotenusaTriangulo
2   Definir catetoA, catetoB, hipotenusa Como Real
3
4   Repetir
5       Escribir "Ingresa el valor del primer Cateto:"
6       Leer catetoA
7       Si catetoA <= 0 Entonces
8           Escribir "Ingreso un valor positivo distinto de 0."
9       FinSi
10  Hasta Que catetoA > 0
11
12  Repetir
13      Escribir "Ingresa el valor del segundo Cateto:"
14      Leer catetoB
15      Si catetoB <= 0 Entonces
16          Escribir "Ingreso un valor positivo distinto de 0."
17      FinSi
18  Hasta Que catetoB > 0
19
20  hipotenusa <- raiz((catetoA*catetoA) + (catetoB*catetoB))
21
22  Escribir "El valor de la hipotenusa es: ", hipotenusa
23
24 FinAlgoritmo
```

Algoritmo 4: Cálculo de la hipotenusa

[↑ Volver al índice](#)

5. Área y volumen de un cilindro

Desarrolle un algoritmo que permita determinar el área y volumen de un cilindro dado su radio (R) y altura (H).

5.1. Algoritmo principal

```
1 Algoritmo AreaVolumenCilindro
2   Definir radio, altura, area, volumen Como Real
3
4   Repetir
5       Escribir "Ingresa el valor del radio del cilindro:"
6       Leer radio
7       Si radio <= 0 Entonces
8           Escribir "El radio debe ser positivo y distinto de
9               0."
10          FinSi
11  Hasta Que radio > 0
12
13  Repetir
14      Escribir "Ingresa el valor de la altura del cilindro:"
15      Leer altura
16      Si altura <= 0 Entonces
17          Escribir "La altura debe ser positiva y distinta de
18              0."
19      FinSi
20  Hasta Que altura > 0
21
22  area <- (2*PI*radio*radio) + (2*PI*radio*altura) // area
23      lateral + area de c/tapa
24  volumen <- (PI*radio*radio*altura)
25
26  Escribir "El area del Cilindro es: ", area
27  Escribir "El volumen del cilindro es: ", volumen
28
29 FinAlgoritmo
```

Algoritmo 5: Área y volumen de un cilindro

[↑ Volver al índice](#)

6. Determinar si un número es par o impar

Desarrolle un algoritmo que permita leer un valor cualquiera N y escriba si dicho número es par o impar.

6.1. Algoritmo principal

```
1 Algoritmo ParImpar
2   Definir n Como Entero
3
4   Escribir "Ingresa un numero:"
5   Leer n
6
7   Si (n MOD 2 = 0) Entonces
8       Escribir n, " es par."
9   SiNo
10      Escribir n, " es impar."
11  FinSi
12
13 FinAlgoritmo
```

Algoritmo 6: Par o impar

[↑ Volver al índice](#)

7. Conversión de calificaciones numéricas a letras

Desarrolle un algoritmo que permita convertir calificaciones numéricas, según la siguiente tabla: $A = 19$ y 20 , $B = 16, 17$ y 18 , $C = 13, 14$ y 15 , $D = 10, 11$ y 12 , $E = 1$ hasta el 9 . Se asume que la nota está comprendida entre 1 y 20 .

7.1. Algoritmo principal (Si-Entonces)

```
1 Algoritmo Calificaciones
2   Definir calificacion Como Real
3
4   Escribir "Ingresa la calificacion:"
5   Leer calificacion
6
7   Si (calificacion >= 19 Y calificacion <= 20) Entonces
8       Escribir "A"
9   FinSi
10
11  Si (calificacion >= 16 Y calificacion <= 18) Entonces
12      Escribir "B"
13  FinSi
14
15  Si (calificacion >= 13 Y calificacion <= 15) Entonces
16      Escribir "C"
17  FinSi
18
19  Si (calificacion >= 10 Y calificacion <= 12) Entonces
20      Escribir "D"
21  FinSi
22
23  Si (calificacion >= 1 Y calificacion <= 9) Entonces
24      Escribir "E"
25  FinSi
26
27 FinAlgoritmo
```

Algoritmo 7: Conversión de calificaciones con Si-Entonces

[↑ Volver al índice](#)

8. Ordenar dos números de menor a mayor

Desarrolle un algoritmo que permita leer dos números y ordenarlos de menor a mayor, si es el caso.

8.1. Algoritmo principal

```
1 Algoritmo OrdenadorNumeros
2   Definir n1, n2 Como Real
3
4   Escribir "Ingresa el primer numero:"
5   Leer n1
6
7   Escribir "Ingresa el segundo numero:"
8   Leer n2
9
10  Si (n1 >= n2) Entonces
11      Escribir n2 ", " n1
12  SiNo
13      Escribir n1 ", " n2
14  FinSi
15
16 FinAlgoritmo
```

Algoritmo 8: Ordenar dos números

[↑ Volver al índice](#)

9. Determinar si un número es primo

Desarrolle un algoritmo que permita leer un valor entero positivo N y determinar si es primo o no.

9.1. Algoritmo principal

```
1  Algoritmo EsPrimo
2      Definir n, i Como Entero
3      Definir esPrimo Como Logico
4
5      Repetir
6          Escribir "Ingresa un numero. Debe ser mayor a 1:"
7          Leer n
8          Si (n <= 1) Entonces
9              Escribir "Numero invalido. Intente nuevamente."
10         FinSi
11     Hasta Que n > 1
12
13     esPrimo <- Verdadero
14     i <- 2
15
16     Mientras i <= n / 2 Y esPrimo Hacer
17         Si (n MOD i = 0) Entonces
18             esPrimo <- Falso
19         FinSi
20         i <- i + 1
21     FinMientras
22
23     Si esPrimo Entonces
24         Escribir "Es primo."
25     SiNo
26         Escribir "No es primo."
27     FinSi
28
29 FinAlgoritmo
```

Algoritmo 9: Determinar si N es primo

[↑ Volver al índice](#)

10. Corrección de errores: cálculo de estacionamiento

Realice un algoritmo que calcule el monto a pagar por el servicio de estacionamiento, teniendo en cuenta que por la primera hora de estadía se tiene una tarifa de 200 pesos y las restantes tienen un costo de 100 pesos. Se tiene como datos: hora de entrada, hora de salida, iniciada una hora se contabiliza como hora total.

10.1. Algoritmo corregido

```
1 Algoritmo Estacionamiento
2   Definir horaEntrada, horaSalida Como Entero
3   Definir tarifaPrimeraHora, tarifaPorHora, tarifa Como Real
4   tarifaPrimeraHora <- 200
5   tarifaPorHora <- 100
6
7   Repetir
8       Escribir "Ingresa la hora de entrada (0-23):"
9       Leer horaEntrada
10      Si (horaEntrada < 0 O horaEntrada > 23) Entonces
11          Escribir "Error: la hora debe estar entre 0 y 23."
12      FinSi
13  Hasta Que horaEntrada >= 0 Y horaEntrada <= 23
14
15  Repetir
16      Escribir "Ingresa la hora de salida (0-23):"
17      Leer horaSalida
18      Si (horaSalida < horaEntrada O horaSalida > 23) Entonces
19          Escribir "Error: la hora de salida debe ser mayor o
20              igual a la de entrada y estar entre 0 y 23."
21      FinSi
22  Hasta Que horaSalida >= horaEntrada Y horaSalida <= 23
23  // Nota: Se asume que la entrada y la salida ocurren el
24  mismo día.
25
26  Si (horaSalida - horaEntrada <= 1) Entonces
27      tarifa <- tarifaPrimeraHora
28  SiNo
29      tarifa <- tarifaPrimeraHora + tarifaPorHora *
30          ((horaSalida - horaEntrada) - 1)
31  FinSi
32
33  Escribir "La tarifa total es: $", tarifa
34 FinAlgoritmo
```

Algoritmo 10: Cálculo del monto de estacionamiento

[↑ Volver al índice](#)

11. Conversión de velocidad: km/h a m/s

Realice un algoritmo que a partir de proporcionarle la velocidad de un auto-móvil, expresada en kilómetros por hora, proporcione la velocidad en metros por segundo.

11.1. Algoritmo principal

```
1 Algoritmo ConversorVelocidad
2   Definir velocidadEnKMH, velocidadEnMS Como Real
3
4   Escribir "Ingresa la velocidad en Km/h:"
5   Leer velocidadEnKMH
6
7   velocidadEnMS <- velocidadEnKMH * 1000 / (60*60)
8
9   Escribir "La velocidad en m/s es:", velocidadEnMS
10 FinAlgoritmo
```

Algoritmo 11: Conversión de km/h a m/s

[↑ Volver al índice](#)

12. Cálculo del promedio de notas

Desarrolle un algoritmo que permita calcular el promedio de notas; finaliza cuando $N = 0$.

12.1. Algoritmo principal

```
1 Algoritmo PromedioNotas
2   Definir nota Como Real
3   Definir suma, promedio Como Real
4   Definir contador Como Entero
5
6   suma <- 0
7   contador <- 0
8
9   Escribir "Ingresa una nota (ingresa 0 para finalizar):"
10  Leer nota
11
12  Mientras nota <> 0 Hacer
13      suma <- suma + nota
14      contador <- contador + 1
15
16      Escribir "Ingresa una nota (ingresa 0 para finalizar):"
17      Leer nota
18  FinMientras
19
20  Si contador > 0 Entonces
21      promedio <- suma / contador
22      Escribir "El promedio de las notas es: ", promedio
23  SinO
24      Escribir "No se ingresaron notas para calcular el
25          promedio."
26  FinSi
27 FinAlgoritmo
```

Algoritmo 12: Promedio de notas

[↑ Volver al índice](#)

13. Nómina de obreros - Constructora VILAZ S.A.

Desarrolle un algoritmo para la empresa Constructora VILAZ S.A., que le permita calcular e imprimir la nómina para su cancelación a un total de 50 obreros calificados a quienes debe cancelar por horas trabajadas. La hora trabajada se pautó en 1500 pesos.

13.1. Algoritmo principal

```
1 Algoritmo NominaObreros
2   Definir numObrero Como Entero
3   Definir horasTrabajadas Como Real
4   Definir valorHora Como Real
5   Definir pagoObrero, totalNomina Como Real
6
7   valorHora <- 1500
8   totalNomina <- 0
9
10  Para numObrero <- 1 Hasta 50 Con Paso 1 Hacer
11      Escribir "--- Obrero Nro: ", numObrero, " ---"
12      Escribir "Ingrese horas trabajadas: "
13      Leer horasTrabajadas
14
15      pagoObrero <- horasTrabajadas * valorHora
16      totalNomina <- totalNomina + pagoObrero
17
18      Escribir "Pago a cancelar: $", pagoObrero
19  FinPara
20
21  Escribir "==== RESUMEN GENERAL ====="
22  Escribir "Total nomina a cancelar (50 obreros): $",
23      totalNomina
24 FinAlgoritmo
```

Algoritmo 13: Nómina de 50 obreros

[↑ Volver al índice](#)

14. Simulación de caja registradora

Desarrolle un algoritmo que funcione como caja registradora.

14.1. Algoritmo principal

```
1 Algoritmo CajaRegistradora
2   Definir precio Como Real
3   Definir total, montoPagado, vuelto Como Real
4   Definir cantidadProductos Como Entero
5
6   total <- 0
7   cantidadProductos <- 0
8
9   Escribir "==== CAJA REGISTRADORA ====="
10  Escribir "Ingresa el precio del producto (ingresa 0 para
      finalizar):"
11  Leer precio
12
13  Mientras precio <> 0 Hacer
14      total <- total + precio
15      cantidadProductos <- cantidadProductos + 1
16
17      Escribir "Producto agregado. Precio: $", precio
18      Escribir "Ingrese el precio del producto (ingrese 0 para
          finalizar):"
19      Leer precio
20  FinMientras
21
22  Si cantidadProductos > 0 Entonces
23      Escribir "==== TICKET DE COMPRA ====="
24      Escribir "Cantidad de productos: ", cantidadProductos
25      Escribir "Total a pagar: $", total
26
27      Escribir "Ingrese el monto con el que paga el cliente:"
28      Leer montoPagado
29
30      Mientras montoPagado < total Hacer
31          Escribir "El monto ingresado es insuficiente.
          Ingrese un monto mayor o igual al total:"
32          Leer montoPagado
33      FinMientras
34
35      vuelto <- montoPagado - total
```

```
36      Escribir "Monto pagado: $", montoPagado
37      Escribir "Vuelto a entregar: $", vuelto
39  SinO
40      Escribir "No se registraron productos en esta compra."
41  FinSi
42
43 FinAlgoritmo
```

Algoritmo 14: Caja registradora

[↑ Volver al índice](#)

15. Conversión de días a años, meses, semanas y días

Desarrolle un algoritmo que permita determinar a partir de un número de días, ingresado por pantalla, cuántos años, meses, semanas y días constituyen el número de días proporcionado, utilizando la estructura Mientras o While.

15.1. Algoritmo principal (Mientras)

```
1  Algoritmo ConversionDias
2      Definir diasIngresados Como Entero
3      Definir anios, meses, semanas, dias Como Entero
4
5      anios <- 0
6      meses <- 0
7      semanas <- 0
8      dias <- 0
9
10     Escribir "Ingresa la cantidad de dias:"
11     Leer diasIngresados
12     dias <- diasIngresados
13
14     Mientras dias >= 365 Hacer
15         anios <- anios + 1
16         dias <- dias - 365
17     FinMientras
18
19     Mientras dias >= 30 Hacer
20         meses <- meses + 1
21         dias <- dias - 30
22     FinMientras
23
24     Mientras dias >= 7 Hacer
25         semanas <- semanas + 1
26         dias <- dias - 7
27     FinMientras
28
29     Escribir "El numero de dias ingresado (", diasIngresados, ")
30         equivale a:"
31     Escribir "Años: ", anios
32     Escribir "Meses: ", meses
33     Escribir "Semanas: ", semanas
34     Escribir "Dias: ", dias
35 FinAlgoritmo
```

Algoritmo 15: Conversión de días con Mientras

[↑ Volver al índice](#)